Soluzioni del compito di Metodi Matematici e Calcolo per Ingegneria del 19 luglio 2005

Enrico Bertolazzi

Trasformata di Laplace

```
> restart:
  with(inttrans) :
```

Data la seguente funzione

```
> f := t -> cos(t)+sin(t) ;
```

$$f := t \to \cos(t) + \sin(t)$$

Usando le regole di trasformazione calcolare le trasformate delle funzioni

```
> f1 := t -> f(3*t) ;
  f2 := t -> f(2*t)*exp(-t) ;
  f3 := t -> D(D(f))(t) ;
```

$$f1 := t \to f(3\,t)$$

$$f2 := t \to f(2\,t)\,\mathbf{e}^{(-t)}$$

$$f3 := t \to \mathrm{D}(\mathrm{D}(f))(t)$$

Trasformate con le primitive Maple

```
> laplace(f(t),t,s);
  laplace(f1(t),t,s);
  laplace(f2(t),t,s);
  laplace(f3(t),t,s);
```

$$\frac{s+1}{s^2+1}$$

$$\frac{s+3}{s^2+9}$$

$$\frac{s+3}{s^2+2\,s+5}$$

$$-\frac{s+1}{s^2+1}$$

Soluzione di ODE con Laplace

```
> restart:
  with(inttrans) :
```

Data la seguente equazione differenziale

```
> ode := diff(y(x),x,x)-y(x)=sin(-x) ;
```

$$ode := \left(\frac{d^2}{dx^2} y(x)\right) - y(x) = -\sin(x)$$

Con dato iniziale

```
> y0, yp0 := 1,-1 ;
```

$$y0, yp0 := 1, -1$$

Calcolare le soluzioni con le trasformate di Laplace.

Trasformo la equazione differenziale con la trasformata di Laplace

```
> sode := laplace(ode,x,s) ;
```

$$sode := s^2\, laplace(y(x), x, s) - \mathrm{D}(y)(0) - s\, y(0) - laplace(y(x), x, s) = -\frac{1}{s^2+1}$$

Risolvo la equazione per y(s)

```
> lode := isolate(sode,laplace(y(x),x,s));
```

$$lode := laplace(y(x), x, s) = \frac{-\frac{1}{s^2+1} + \mathrm{D}(y)(0) + s\, y(0)}{s^2-1}$$

Applico le condizioni iniziali ottenendo y(s)

```
> ly := subs( y(0)=y0,D(y)(0)=yp0,D(D(y))(0)=ypp0,rhs(lode)) ;
```

$$ly := \frac{-\frac{1}{s^2+1} - 1 + s}{s^2-1}$$

Espansione in fratti semplici

```
> convert(ly, parfrac, s);
```

$$\frac{5}{4\,(s+1)} - \frac{1}{4\,(s-1)} + \frac{1}{2\,(s^2+1)}$$

Antitrasformo per ottenere la equazione y(x)

```
> res := invlaplace(ly,s,t) ;
```

$$res := -\frac{1}{4}\mathrm{e}^{t} + \frac{1}{2}\sin(t) + \frac{5}{4}\mathrm{e}^{(-t)}$$

Soluzione di un sistema di ODE con Laplace

```
> restart:
  with(inttrans) :
```

Dato il seguente sistema di equazioni differenziali

```
> ode1 := diff(y(x),x)-2*diff(z(x),x)=cos(x) ;
  ode2 := -2*diff(y(x),x)+diff(z(x),x)=sin(x) ;
```

$$ode1 := \left(\frac{d}{dx}y(x)\right) - 2\left(\frac{d}{dx}z(x)\right) = \cos(x)$$

$$ode2 := -2\left(\frac{d}{dx}y(x)\right) + \left(\frac{d}{dx}z(x)\right) = \sin(x)$$

Con dato iniziale

```
> y0, z0 := 0,0 ;
```

$$y0, z0 := 0, 0$$

Calcolare le soluzioni con le trasformate di Laplace.

Trasformo le equazioni differenziale con la trasformata di Laplace

```
> sode1 := laplace(ode1,x,s) ;
  sode2 := laplace(ode2,x,s) ;
```

$$sode1 := s\,laplace(y(x), x, s) - y(0) - 2\,s\,laplace(z(x), x, s) + 2\,z(0) = \frac{s}{s^2+1}$$

$$sode2 := -2\,s\,laplace(y(x), x, s) + 2\,y(0) + s\,laplace(z(x), x, s) - z(0) = \frac{1}{s^2+1}$$

Risolvo la equazione per y(s), z(s)

```
> RES := solve({sode1,sode2},{laplace(y(x),x,s),laplace(z(x),x,s)});
```

$$RES := \left\{ laplace(y(x), x, s) = \frac{-s + 3\,y(0)\,s^2 + 3\,y(0) - 2}{3\,s\,(s^2+1)},\ laplace(z(x), x, s) = \frac{-2\,s - 1 + 3\,z(0)\,s^2 + 3\,z(0)}{3\,s\,(s^2+1)} \right\}$$

Applico le condizioni iniziali ottenendo y(s), z(s)

```
> SOL := subs(RES,y(0)=y0,z(0)=z0,<laplace(y(x),x,s),laplace(z(x),x,s)>);
```

$$SOL := \begin{bmatrix} \dfrac{-s-2}{3\,s\,(s^2+1)} \\ \dfrac{-2\,s-1}{3\,s\,(s^2+1)} \end{bmatrix}$$

Antitrasformo per ottenere y(x), z(x)

```
> yy := invlaplace(SOL[1],s,x) ;
  zz := invlaplace(SOL[2],s,x) ;
```

$$yy := \frac{2}{3}\cos(x) - \frac{1}{3}\sin(x) - \frac{2}{3}$$

$$zz := \frac{1}{3}\cos(x) - \frac{2}{3}\sin(x) - \frac{1}{3}$$

Espansione in fratti semplici per controllo

```
> convert(SOL[1], parfrac, s);
  convert(SOL[2], parfrac, s);
```

$$-\frac{2}{3\,s} + \frac{-1+2\,s}{3\,(s^2+1)}$$

$$-\frac{1}{3\,s} + \frac{-2+s}{3\,(s^2+1)}$$

Soluzione di ricorrenza con trasformata zeta

```
> restart:
```

Risolvere la seguente ricorrenza

```
> RIC := f(n+2) = 2*f(n+1) + 3*f(n) +1  ;
```

$$RIC := f(n+2) = 2\,f(n+1) + 3\,f(n) + 1$$

Con dato iniziale

```
> INI := f(0)=0,f(1)=0 ;
```

$$INI := f(0) = 0, f(1) = 0$$

Usando le primitive si maple:

```
> rsolve({RIC,INI}, f(k));
```

$$-\frac{1}{4} + \frac{1}{8}(-1)^k + \frac{1}{8}3^k$$

```
> simplify(%);
```

$$-\frac{1}{4} + \frac{1}{8}(-1)^k + \frac{1}{8}3^k$$

Usando la Z-trasformata (occhio negli appunti abbiamo Z(f) = sum(f(n)*z^n mentre maple usa Z(f) = sum(f(n)*w^{-n}
Quindi per confrontare le trasformate bisogna trasfromare w -> 1/z;

```
> zRIC := ztrans(RIC,n,w):
  simplify(subs( ztrans(f(n),n,w)=f(z), w=1/z, zRIC)) ;
```

$$\frac{f(z) - f(0) - f(1)\,z}{z^2} = \frac{-2\,f(z) - f(z)\,z + 2\,f(0) - 2\,f(0)\,z + 3\,f(z)\,z^2 - z}{z\,(-1+z)}$$

Ricavo f(w) [f(1/z)]

```
> zRICrhs := isolate(zRIC,ztrans(f(n),n,w)):
```

```
simplify(subs( ztrans(f(n),n,w)=f(z), w=1/z, zRICrhs)) ;
```

$$f(z)=\frac{f(0)-3\,f(0)\,z+f(1)\,z-f(1)\,z^2+2\,f(0)\,z^2+z^2}{(-1+z)\,(-1+2\,z+3\,z^2)}$$

Applico le condizioni iniziali

```
> zRICrhsINI := subs(INI,zRICrhs):
  simplify(subs( ztrans(f(n),n,w)=f(z), w=1/z, zRICrhsINI)) ;
```

$$f(z)=\frac{z^2}{(-1+z)\,(-1+2\,z+3\,z^2)}$$

Conversione in fratti semplici

```
> convert(%, parfrac);
```

$$f(z)=\frac{1}{4\,(-1+z)}+\frac{1}{8\,(z+1)}-\frac{1}{8\,(3\,z-1)}$$

Inversione della Z-trasformata

```
> invztrans(zRICrhsINI,w,k) ;
```

$$f(k)=-\frac{1}{4}+\frac{1}{8}\,3^k+\frac{1}{8}\,(-1)^k$$

Soluzione di un sistema non lineare con Newton

```
> restart:
  with(VectorCalculus):
Warning, the assigned names `<,>` and `<|>` now have a global binding

Warning, these protected names have been redefined and unprotected: `*`,
`+`, `.`, D, Vector, diff, int, limit, series
```

Sistema non lineare

```
> f := y - x + 1 ;
  g := exp(x) - exp(-y) ;
```

$$f:=y-x+1$$

$$g:=\mathbf{e}^{x}-\mathbf{e}^{(-y)}$$

Soluzione esatta

```
> solve({f,g},{x,y}) ;
```

$$\left\{y=\frac{-1}{2},\,x=\frac{1}{2}\right\}$$

Matrice Jacobiano

```
> J := Jacobian([f,g],[x,y]) ;
```

$$J := \begin{bmatrix} -1 & 1 \\ e^{x} & e^{(-y)} \end{bmatrix}$$

Schema di Newton

```
> Newton_update := <x,y>-J^(-1).<f,g> ;
```

$$Newton\_update := \left(x + \frac{e^{(-y)}\,(y - x + 1)}{e^{(-y)} + e^{x}} - \frac{e^{x} - e^{(-y)}}{e^{(-y)} + e^{x}} \right) e_x + \left(y - \frac{e^{x}\,(y - x + 1)}{e^{(-y)} + e^{x}} - \frac{e^{x} - e^{(-y)}}{e^{(-y)} + e^{x}} \right) e_y$$

Schema di Newton per questo sistema non lineare

```
> x[k+1]=simplify(subs(x=x[k],y=y[k],Newton_update[1])) ;
  y[k+1]=simplify(subs(x=x[k],y=y[k],Newton_update[2])) ;
```

$$x_{k+1} = \frac{x_k\, e^{x_k} + e^{(-y_k)}\, y_k + 2\, e^{(-y_k)} - e^{x_k}}{e^{(-y_k)} + e^{x_k}}$$

$$y_{k+1} = \frac{e^{(-y_k)}\, y_k + x_k\, e^{x_k} - 2\, e^{x_k} + e^{(-y_k)}}{e^{(-y_k)} + e^{x_k}}$$

Tre iterate a partire da (1,2)

```
> x[0],y[0]:= 1,2 ;
```

$$x_0, y_0 := 1, 2$$

Prima iterata

```
> x[1] := evalf(subs(x=x[0],y=y[0],Newton_update[1])) ;
  y[1] := evalf(subs(x=x[0],y=y[0],Newton_update[2])) ;
```

$$x_1 := 0.1897034923$$

$$y_1 := -0.8102965077$$

Seconda iterata

```
> x[2] := evalf(subs(x=x[1],y=y[1],Newton_update[1])) ;
  y[2] := evalf(subs(x=x[1],y=y[1],Newton_update[2])) ;
```

$$x_2 := 0.4904103097$$

$$y_2 := -0.5095896903$$

Terza iterata

```
> x[3] := evalf(subs(x=x[2],y=y[2],Newton_update[1])) ;
  y[3] := evalf(subs(x=x[2],y=y[2],Newton_update[2])) ;
```

$$x_3 := 0.4999997059$$

$$y_3 := -0.5000002941$$

Problema di Minimo Vincolato

```
> restart:
  with(LinearAlgebra):
  with(Optimization):
  with(VectorCalculus):
Warning, the names `&x`, CrossProduct and DotProduct have been rebound

Warning, the assigned names `<,>` and `<|>` now have a global binding

Warning, these protected names have been redefined and unprotected: `*`,
`+`, `.`, D, Vector, diff, int, limit, series
```

Minimizzare la seguente funzione

```
> f := x+y^2+z^3;
```

$$f := x + y^2 + z^3$$

Soggetta ai seguenti vincoli

```
> v := [x=3,x-z=0] ;
```

$$v := [x = 3, x - z = 0]$$

Soluzione con le primitive Maple

```
> Minimize(f, v );
```

$$[30., [y = 1.35525271560688054\ 10^{-20}, z = 3., x = 3.]]$$

Uso dei moltiplicatori di Lagrange

```
> v1 := lhs(v[1])-rhs(v[1]) ;
  v2 := lhs(v[2])-rhs(v[2]) ;
```

$$v1 := x - 3$$

$$v2 := x - z$$

```
> g := f - lambda*v1 - mu*v2 ;
```

$$g := x + y^2 + z^3 - \lambda (x - 3) - \mu (x - z)$$

Sistema non lineare da risolvere

```
> F := Gradient(g,[x,y,z,lambda,mu]) ;
```

$$F := (1 - \lambda - \mu)\, \bar{e}_x + 2\, y\, \bar{e}_y + (3\, z^2 + \mu)\, \bar{e}_z + (-x + 3)\, \bar{e}_\lambda + (-x + z)\, \bar{e}_\mu$$

```
> RES := solve({F[1],F[2],F[3],F[4],F[5]},{x,y,z,lambda,mu}) ;
```

$$RES := \{\mu = -27, \lambda = 28, z = 3, x = 3, y = 0\}$$

Controllo proprietà di minimo

```
> Hf  := Hessian(f,[x,y,z]):
```

```
Hv1 := Hessian(v1,[x,y,z]):
Hv2 := Hessian(v2,[x,y,z]):
Hf, Hv1, Hv2 ;
```

$$\begin{bmatrix} 0 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 6z \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

```
> JH := Jacobian([v1,v2],[x,y,z]) ;
  NH := NullSpace(JH) ;
```

$$JH := \begin{bmatrix} 1 & 0 & 0 \\ 1 & 0 & -1 \end{bmatrix}$$

$$NH := \left\{ \begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix} \right\}$$

Controllo minimo/massimo locale primo punto

```
> lambda1 := subs(RES,lambda);
  mu1     := subs(RES,mu);
```

$$\lambda 1 := 28$$

$$\mu 1 := -27$$

```
> Hf1 := simplify(Hf - lambda1. Hv1 - mu1. Hv2) ;
```

$$Hf1 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 6z \end{bmatrix}$$

Cerco nello spazio dei vincoli:

```
> Z1 := subs(RES,op(NH)) ;
```

$$Z1 := \begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix}$$

E' positivo per ogni alpha > 0, quindi è un minimo locale

```
> simplify(Transpose(alpha.Z1).Hf1.(alpha.Z1)) ;
```

$$2\alpha^2$$

```
> subs(RES,f);
```

$$30$$

```
> evalf(%);
```

$$30.$$

```
>
```